Units and IF in Simplex/

Two Modelling Exercises

Units in Simplex

Simplex is unique in its use of units (dimensions)

All model quantities may be assigned a unit

Examples:

- **Volt [V]**
- **Kilogram [kg]**
- **Newton [N]**
- **Metre [m]**
- **Second [s]**

Units in Simplex

Units are written in square brackets: `[m]`

If units are being used, then <u>all</u> values must be given a unit

The unit of simulation time T must then also be declared:

```
USE OF UNITS
   TIMEUNIT = [s]
```

Simplex automatically checks all units

- Incompatible units generate a syntax error!

Units in Simplex

Conversion of units is automatic:

```
DEPENDENT VARIABLES
CONTINUOUS
  Velo (REAL [km/h]) := 0.0 [km/h],
  Dist (REAL [m]   ) := 1.0 [m],
  Time (REAL [s]   ) := 2.0 [s]

DYNAMIC BEHAVIOUR
  ...
  Velo := Dist / Time + 1.0 [km/h];
```

IF-Statements in Simplex

Simplex allows IF - ELSIF - ELSE statements:

```
IF (y < 10)
DO
     f := 0.0;
END
ELSIF (y >= 10) AND (y < 20)
DO
     f := 1.0;
END
ELSE
DO
     f := 2.0;
END
```


Modelling an Electrical Circuit



Modelling an Electrical Circuit

Electrical circuits are a good example for ODE modelling

They illustrate the model-building principle well:

- Find out the quantities that are moved / conserved
- Model the behaviour of the individual components
- Build the conservation equation



Modelling an Electrical Circuit

Each component is characterised by

- The voltage V drop across it
- The current I flowing through it

Physical principle of the closed loop (Kirchhoff's law):

The sum of the voltages across all elements is equal to zero

- $E = V_L + V_C + V_R$



Modelling an Electrical Circuit

State variables:

Current I [Ampere]

Charge Q [Coulomb]



Current = rate of change of Charge

Electrical Components: Resistor

A resistor is made of a material with low conductance

It has a Resistance R [Ohm]

The resistor's behaviour is described by Ohm's law:

$$V_R = R \cdot I \text{ [Volts]}$$



Electrical Components: Capacitor

A capacitor consists of two closely-spaced conducting plates

It stores energy in an electrostatic field between the plates

A capacitor has

- a Capacitance C [Farad]
- a Charge Q [Coulomb]

The capacitor is described by



$$V_C = Q / C \text{ [Volts]}$$

$$I = dQ / dt \text{ [Ampere]}$$

Electrical Components: Inductor

The inductor is built as a wire coil

It stores energy in a magnetic field

It has an Inductance L [Henry]

The inductor is described by

$$V_L = L \cdot dI / dt \text{ [Volts]}$$



The Declaration Part

```
BASIC COMPONENT Circuit

USE OF UNITS
  TIMEUNIT = [s]

DECLARATION OF ELEMENTS
  CONSTANTS
    L (REAL [H])   := 1.0E-3  [H],   # Inductance
    C (REAL [F])   := 1.0E-6  [F],   # Capacitance
    R (REAL [Ohm]) := 1.0E+1  [Ohm]  # Resistance

  STATE VARIABLES
  CONTINUOUS
    I (REAL [A])   := 0.0  [A],      # Current in circuit
    Q (REAL [C])   := 0.0  [C]       # Charge on capacitor

  DEPENDENT VARIABLES
  CONTINUOUS
    E (REAL [V])   := 0.0  [V]       # Voltage applied
```


Equations of State

We have:

$$I = dQ / dt$$



$$E = V_L + V_C + V_R$$

$$= L \cdot dI / dt + Q/C + R \cdot I$$

Equations of State

We have:

$$I = dQ / dt \longrightarrow Q' = I$$

$$E = V_L + V_C + V_R$$

$$= L \cdot dI / dt + Q/C + R \cdot I \longrightarrow I' = (E - R \cdot I - Q/C)/L$$

The Dynamic Behaviour

```
DYNAMIC BEHAVIOUR

DIFFERENTIAL EQUATIONS
  Q' := I;
  I' := (E - R*I - Q/C)/L;
END

E :=  1.0 [V];

END OF Circuit
```

Simulation Results

Simulation results:



The Bungee Jumper



The Bungee Jumper

Definition of relevant quantities:

- Length of rope: *l* [m]
- Spring constant of rope: *k* [N/m]
- Damping constant of rope: *D* [N*s/m]
- Distance of jumper below bridge: *y* [m]
- Downward velocity of jumper: *v* [m/s]
- Mass of jumper: *m* [kg]
- Acceleration due to gravity: *g* [m/s^2]

Springs and Dampers

When taut, the rope exerts two downward (!) forces:

1) proportional to its length of extension:

$$\text{Force} = -\,k \cdot \text{extension}$$

2) proportional to its speed of extension:

$$\text{Force} = -\,D \cdot \text{rate of extension}$$

The Downward Forces

Let F be the rope's *downward* force on the jumper

When $y < l$, the rope is slack:

$$F = 0$$

When $y > l$, the rope is taut and pulls up:

$$F = -k \cdot (y - l) - D \cdot v$$

Equations

We need equations for position y and velocity v

Position:

- Definition of speed: $v = dy/dt$

Speed:

- Definition of acceleration: $a = dv/dt$
- Newton's Law: acceleration = force / mass

Result:

$$\frac{dy}{dt} = v \qquad \frac{dv}{dt} = g + F/m$$

The Constants

```
BASIC COMPONENT  Bungee

USE OF UNITS
   TIMEUNIT = [s]

DECLARATION OF ELEMENTS
   CONSTANTS
      m    (REAL [kg])     :=  60.0   [kg],
      D    (REAL [N*s/m]) :=  10.0   [N*s/m],
      k    (REAL [N/m])    :=  50.0   [N/m],
      g    (REAL [m/s^2]) :=   9.81  [m/s^2],
      l    (REAL [m])      :=  20.0   [m]
```


The Variables

```
STATE VARIABLES
   CONTINUOUS
      y (REAL [m])      :=   0.0   [m],
      v (REAL [m/s])    :=   0.0   [m/s]

   DEPENDENT VARIABLES
   CONTINUOUS
      f (REAL [N])      :=   0.0   [N]
```

The Dynamic Behaviour

```
DYNAMIC BEHAVIOUR

   DIFFERENTIAL EQUATIONS
      y'  := v;
      v'  := g + f/m;
   END

   IF (y < l)
   DO
      f   := 0.0 [N];
   END
   ELSE DO
      f   := - k*(y-l) - D*v;
   END

END OF Bungee
```


Simulation Result

Simulation result for y :